Surface Area and Volume via Vector Calculus

We will use Mathematica to find the surface area and volume of an oblate spheroid which has the form $\frac{x^2}{a^2} + \frac{y^2}{a^2} + \frac{z^2}{c^2} = 1$ where $a > c > 0$. We being by parametrizing the surface using a simple variation of spherical coordinates.

```
In[12]:= Ell[θ_, ϕ_] := {a * Cos[θ] * Sin[ϕ], a * Sin[θ] * Sin[ϕ], c * Cos[ϕ]}
```

Notice that the angle θ ranges from 0 to 2π while ϕ ranges from 0 to π.

Surface Area:

For surface area, we need to integrate 1 over the surface. For a scalar surface integral, the order we cross the partial derivative vectors is irrelevant. In order to make full use of the various simplification routines in Mathematica, we compute the cross product and then manually add up the squares of the individual components.

```
In[13]:= cross = Cross[D[Ell[θ, ϕ], θ], D[Ell[θ, ϕ], ϕ]];

In[17]:= normCross =
  FullSimplify[Sqrt[cross[[1]]^2 + cross[[2]]^2 + cross[[3]]^2], Assumptions → a > c && c > 0]
```

Out[17]=

$$\frac{a\sqrt{\left(a^2 + c^2 + (a - c)(a + c)\,\mathrm{Cos}[2\,\phi]\right)\mathrm{Sin}[\phi]^2}}{\sqrt{2}}$$

```
In[18]:= Integrate[normCross, {θ, 0, 2 * Pi}, {ϕ, 0, Pi}]
```

Out[18]=

$$\frac{1}{2}\,a\,\pi\left(\frac{\sqrt{c^2}\left(2\,a^2\sqrt{1 - \frac{a^2}{c^2}} + \sqrt{\frac{a^2}{c^2}}\;c^2\,\pi - 2\sqrt{\frac{a^2}{c^2}}\;c^2\,\mathrm{ArcSin}\left[\sqrt{\frac{a^2}{c^2}}\right]\right)}{\sqrt{a^2\left(-a^2 + c^2\right)}} + \frac{2\,a^2\sqrt{-1 + \frac{c^2}{a^2}} + c^2\,\pi - 2\,c^2\,\mathrm{ArcSin}\left[\frac{1}{\sqrt{\frac{c^2}{a^2}}}\right]}{\sqrt{a^2}\;\sqrt{-1 + \frac{c^2}{a^2}}}\right)$$

Volume:

To compute the volume, we use the divergence theorem in the form $\text{vol}(S) = \frac{1}{3}\iint_S \langle x, y, z\rangle \cdot d\mathbf{S}$ where we must be sure to use the outward pointing normal. For spherical coordinates, that means the ϕ-partial crossed with the θ-partial. Notice that the vector ⟨x, y, z⟩ is just the position vector pointing to a point on our spheroid.

In[21]:=
```
integrand = FullSimplify[
  Dot[Ell[θ, ϕ], Cross[D[Ell[θ, ϕ], ϕ], D[Ell[θ, ϕ], θ]]], Assumptions → a > c && c > 0]
```

Out[21]= $a^2 c \, \text{Sin}[\phi]$

In[23]:=
```
Integrate[integrand, {θ, 0, 2 * Pi}, {ϕ, 0, Pi}] / 3
```

Out[23]= $\frac{4}{3} a^2 c \pi$